

APPLICATION OF MOSEL-2 LANGUAGE IN PERFORMANCE AND MODELING OF CELLULAR WIRELESS NETWORKS

AYMEN I. ZREIKAT

Department of Information Technology, Mutah University, Mutah, Karak, Jordan, P.O. Box 7, (zip code:61710)
e-mail: siayzrei@yahoo.com

KEYWORDS

Application of MOSEL-2, Cellular Wireless Networks, Performance Analysis and Modeling, IGL.

ABSTRACT

Today's cellular wireless networks must meet increasing challenges of handling a larger demand for service without a loss of quality, by maximizing the spectral efficiency of the network. Therefore, performance evaluation and modeling of cellular wireless networks is considered to be an important issue to overcome the problem of limited resources of the network. MOSEL-2 (MOdelling Specification and Evaluation Language) offers the ability to evaluate complex systems in a straight, simple and very friendly environment. The old version of MOSEL offers only performance evaluation of complex systems with exponential distribution. However, by the new version of MOSEL (i.e. MOSEL2) with new constructs, it is possible to handle other behaviors with non-exponential distribution. In this paper, the application of MOSEL-2 in cellular wireless networks with mix service is presented. The main objective of the call admission control algorithm and the analysis is to obtain lower handover blocking probability over the new call blocking probability for both voice and data connections, which leads to minimum grade-of-service (i.e. < 0.01) over the whole cell. The numerical analysis of the suggested model with the associated interesting performance measures proves the effectiveness of this simulation in describing and solving this type of systems.

1. INTRODUCTION

1.1 Application of MOSEL-2 in the literature

The generation of mobile networks are evolving very fast starting from 1G, cellular concept based on the analog technology (i.e. only voice), then moving from analog to digital technology (i.e. 2G, GSM and EDGE, providing users with voice and data). The third generation supports broadband voice, data and multimedia services. Performance analysis of multi-service 3G networks plays a major role for mobile network providers, because of the W-CDMA technique used in these systems, which leads to an interference limited systems with a dynamic cell capacity and load dependent cell coverage (Smida et al 2002; Heiska et al 2002).

Performance, reliability modeling and evaluation plays an important role in the design, development, testing, and maintenance of many complex systems in different

applications such as: communication systems, manufacturing systems, computer networks and many other. It has been an extensive use of the old version MOSEL (Al-Begain et al 2001) and the modified version of MOSEL with new constructs, MOSEL-2 (Wuechner 2003) ,in the literature in various applications. The old version of MOSEL has been used intensively for many years and most of these applications are in queueing networks, some of them are in (Zreikat and Bolch 2007; Al-Begain et al 2003; Zreikat et al 2003; Wüchner et al 2007;Gunter et al 2006; Wüchner et al 2005). However, MOSEL-2 language is also used in different applications to model 2G and 3G of mobile networks, some of these examples are in (Zreikat et al 2008; Wüchner et al 2004; Barner and Bolch 2003). Moreover, new work has been done recently to use MOSEL-2 in modeling and evaluation of the 4G of mobile networks (i.e. WIMAX/WIFI or even LTE) (Zreikat 2011). Performance and modeling of one cell of cellular wireless networks with mix service is presented in this paper to show the effectiveness of MOSEL-2 language to describe and solve different types of systems, especially in mobile communications applications.

The paper is organized as follows. In Section 1.2, a description of MOSEL-2 language and environment is given. Modeling of cellular wireless networks is presented in Section 2. In Section 2.1, the modeling assumptions are presented and in Section 2.2, the call admission control algorithm is presented, whereas in Section 2.3, the model solution by MOSEL-2 is presented. The numerical results are presented in Section 3, and finally followed by the conclusions and future work in Section 4.

1.2 Description of MOSEL-2 language and environment

The reliability modeling and evaluation process by MOSEL-2 language is described in Figure 1. There are six main steps that can be summarized as:

- Step 1. The real world system is described by the user via generating a high-level system description using the syntax of MOSEL-2 language. However, the following steps are done without the user interaction. The generated file is saved as filename.mos.
- Step 2 & 3: MOSEL-2 translates the model description into a specific tool and the appropriate tool (MOSES, SPNP or TimeNET) is invoked by

MOSEL-2. This can be done by given an option in the command line. The MOSEL-2 environment is called from a shell using the following command line syntax:

```
>mosel2 options input-file.mos
```

The parameter input file is the name of MOSEL-2 file (for example: filename.mos), which has the suffix ".mos". This MOSEL-2 file is read in, parsed and checked for errors. The options are prefixed by a dash "-" followed by a single letter like "-o". Also multiple options can follow s single dash, like "-Ts", which has the same meanings as "-T -s". To see the list and description of all command options of MOSEL-2, call MOSEL-2 with option "-h" as:

```
>mosel2 -h
```

The option "-s" causes MOSEL-2 to start the selected tool, to read the results and create the result file and if the input file contains picture definitions, an igl file will also be generated, which will contain all the figures specified in the picture part of the file in a nice way. There are 3 additional options for MOSEL-2, usually they are written in association with the "-s" option in order to select the specific tool for evaluation: ("-c", for SPNP tool (Buettel 2003; Wüchner 2005)), ("-T", for TimeNET tool (Buettel 2003; Zimmermann 2005)) or "-m" for MOSES tool (Bolch et al 1998).




Figure 1: Modeling and Evaluation process in MOSEL-2

- Steps 4 & 5: The appropriate tool processes its input file in one of the two following ways (this can be done by different command line options in MOSEL-2):
 - Numerical analysis: out of the static model description, the whole state space of the model is generated by the tool according to the semantic rules of its modeling formalism. This semantic model is mapped onto a stochastic process. The stochastic process is solved by one of the standard

numerical solution algorithms, which are part of the tool.

- Simulation: the model is evaluated by the tool without building the whole state space, using discrete event simulation.

The results of the numerical analysis or discrete event simulation are saved in a file with a tool specific structure.

- Step 6: The MOSEL-2 environment parses the tool specific output and generates a textual result file (i.e., filename.res) which contains the values of the performance and reliability measures according to the user description in the optional "RESULT" part of the program. Additionally, if the optional "PICTURE" part is specified, then the graphical representations of the results are generated by IGL utility in a file "filename.igl".

2. MODELLING OF CELLULAR WIRELESS NETWORKS BY MOSEL-2 LANGUAGE

2.1 Modelling Assumptions

- Only one cell is considered in the analysis and the arrival process over the whole cell is assumed to be Poisson process with {on and of} state for the voice calls.
- The number of mobile terminals in the cell is fixed and assumed to be $M_{Ter} = 500$.
- The number of available channels in the cell NO_CH is 64.
- The call duration or holding time is a random variable which is independent and exponentially distributed with mean $1/\mu_v$ for voice = $1/\mu_d = 100$ seconds.
- The call duration for incoming handover to the cell is also assumed to be a random variable which is independent and exponentially distributed with mean $1/\mu_{hv}$ for voice = $1/\mu_{hd} = 80$ seconds.
- The overall traffic over the cell area is λ . Two types of services are assumed in the analysis: voice and data with the following ratios:
Voice = $\lambda * 0.75$, data = $\lambda * 0.25$.
- The maximum transmission rate of the ongoing data connection is assumed to be 16 (i.e. $max_Trans_rate := 16$).
- The minimum transmission rate of the ongoing connection is assumed to be 4 (i.e. $min_Trans_rate := 4$). It is a threshold for accepting new handover calls.
- Transmission rate threshold is the threshold for accepting the new handover calls. It is assumed in the analysis to be a set of values: 4,7,10 and 13.
- Ideal free space propagation for the signal is assumed.
- No mobility is assumed in the analysis of this paper.

2.2 Call Admission Control

- For incoming calls, the priority is given to the voice calls over the data. All incoming voice connections are served normally as long as there are enough free channels in the cell after accepting a new person. One channel is used for a voice call and `max_Trans_rate` (i.e. maximum transmission rate) for the data connection (is given a value 16 in the code, which means that the data connection is considered to be a multiple of one radio channel). Therefore, the condition for accepting a new voice person will be: Number of voice calls after accepting a new connection $< \text{NO_CH} - 1$.
- In the case of incoming data connection, the call is accepted if there is a free channels but after we exclude the transmission threshold value (i.e. `Trans_rate_thr`) according to the following condition: Number of voice calls after accepting a new connection $< \text{NO_CH} - \text{Trans_rate_thr}$ (is given a range of values in the program, 4-13 to be considered for the next analysis). However, the `Trans_rate_thr` should not be less than the minimum threshold value to admit the new connection. The `min_Trans_rate` (i.e. minimum transmission rate) in the program is given the value 4.
- In case of the incoming handover calls from the neighboring cells either voice or data are treated as a normal incoming call to the cell. The call is only rejected if there are no available channels in the cell after accepting the new call (i.e. 1 channel for the voice connection and `max_Trans_rate` for the data connection). This is why we considered minimum threshold value for accepting the handover calls, which is smaller than both the `Trans_rate_thr` and `max_Trans_rate`. Hence, in our model, the priority is given to the handover calls over the new calls as it is easy to reject a new call rather than stopping the ongoing connection.

2.3 The Model solution by MOSEL-2 language

MOSEL-2 model description of one cell cellular wireless networks is in Figure 2. There are line numbers used in Figure 2. However, there are no line numbers in MOSEL-2 code and the numbers in Figure 2 are used only for referencing. MOSEL-2 specification can be divided into six main parts:

1. The optional constant, parameter and enumerator definition part (Lines 1-14). In a "CONST" definition, the variable is given a floating point constant. In a "PARAMETER" definition, the variable is given a set of values, and the model is evaluated at each value. In "ENUM" definition, the variable is a given a set of constants between two brackets "{}".
2. The node definition part (Lines 15-17), where the nodes are defined. Nodes are used to describe the model's state, each node has a certain value ranges from 0 to a maximum value called the capacity of the node.
3. The optional function and condition part (Lines 18-23). Only function is used in Figure 2 with "FUNC". However, MOSEL-2 offers two types of functions: either

the "FUNC", which yields to a numeric value, or the "COND", which is a placeholder of logical expression.

4. The rule part (Lines 24-31), which contains MOSEL-2 rules. Rules in MOSEL-2 is used to describe how the system may change from one state to another.

5. The optional result part (Lines 32-46). The computation of the required performance measures is done in this part.

6. The optional picture part (47-55). IGL (Intermediate Graphical Language) is a user-friendly tool, which is responsible for this part, to generate the graphical representation of the defined performance measures by the user, in a very nice way. IGL utility is associated with MOSEL-2, where the user can edit, modify and prepare the curves in a nice way. Moreover, the curves can be saved into an encapsulated postscript form (i.e. "figure.eps"), which can be easily viewed and modified in different operating systems. The following interesting performance measures are computed and studied in this paper:

- [1] New call blocking probability: is the fraction of calls, from the new call requests, that are rejected (i.e. blocked) due to the shortages of the available channels.
- [2] Handover blocking probability: is the fraction of handover calls that cannot be admitted due to the shortages of the available channels. This will cause the call request to be terminated.
- [3] Grade-of-service for the data or voice service: is defined as a combination of the new call blocking probability and handover probability multiplied by a factor of 10:
Grade-of-service = new call blocking probability + handover probability *10;
The acceptable level of the grade-of-service is assumed to be $\leq 10\%$.
- [4] Cell utilization: is defined as the average used channels in the cell divide by the total number of available channels in the cell.
- [5] Average data rate: the average transmission rate of the data connection.
- [6] Cell Throughput: is defined as the percentage of voice and data connection that have been successfully served in the cell.

3. NUMERICAL RESULTS

The numerical results are generated by the IGL utility, which is associated with MOSEL-2 language. Nine curves are generated (Figures 3-11) for different transmission rate threshold. This threshold is assumed to have a set of values (4, 7, 10 and 13). The main objective of the analysis is to achieve minimum grade-of-service of both voice and data (i.e. < 0.01) in the assumed cell as well as the whole network. MOSEL-2 model evaluation took only 5 seconds, whereas the same model evaluation by the old version of MOSEL took around 35 seconds. In Figure 3, the grade-of-service for the voice calls is presented. It can be seen that a minimum grade of service (< 0.01) can be achieved even at higher rate for threshold values of 4, 7 and 10.

```

//-----CONSTANT AND PARAMETER PART-----
1 CONST NO_CH:= 64; //NUMBER OF CHANNELS
2 CONST M_Ter:= 500; // finite number of mobile terminals in the cell
3 CONST max_Trans_rate := 16;//maximum transmission rate to the ongoing conection
4 CONST min_Trans_rate := 4;//minimum transmission rate to the ongoing conection
5 PARAMETER Trans_rate_thr :=4..13 STEP 3;//transmission rate threshold:
  //(min_Trans_rate <= Trans_rate_thr <= max_Trans_rate)
6 PARAMETER lambda := 0.10..0.50 STEP 0.10;//arrival rate
7 CONST lambda_for_voice := lambda*0.75;
8 CONST lambda_for_data := lambda*0.25;
9 CONST mue_handover_for_voice := 1.0/80;
10 CONST mue_handover_for_data := 1.0/80;
11 CONST mue_for_voice := 1.0/100;
12 CONST mue_for_data := 1.0/100;
13 CONST data_max_capacity := NO_CH/min_Trans_rate;
14 ENUM on_off_queue:={on,off};
//-----NODES PART-----
15 NODE voice_with_queue[on_off_queue] := on;
16 NODE voice_con[NO_CH];
17 NODE data_con[data_max_capacity];
//-----FUNC PART-----
18 FUNC NO_of_occupied_CH := IF (voice_con+data_con*max_Trans_rate < NO_CH) THEN
  voice_con+data_con*max_Trans_rate ELSE NO_CH;//No. of accupied channels
19 FUNC Handover_thr(x,y):= x+y*min_Trans_rate; //threshold for acceptance of
  handover calls
20 FUNC new_call_thr(x,y):= x+y*Trans_rate_thr;//threshold for acceptance of new calls
21 FUNC Actual_rate_data(x,y):= IF (NO_CH-x)/y >max_Trans_rate THEN max_Trans_rate
  ELSE (NO_CH-x)/y;//actual transmission rate for data connection,
22 FUNC h(x,y):=(NO_CH - x)/y;
23 FUNC reward := IF data_con > 0 THEN h(voice_con,data_con) ELIF
  h(voice_con,data_con) > max_Trans_rate THEN max_Trans_rate ELSE 0; //reward function
//-----RULE PART-----
//---NEW ARRIVAL-----
24 IF (new_call_thr(voice_con,data_con) <= NO_CH-1) AND (voice_with_queue == on) FROM
  EXTERN TO voice_con RATE (M_Ter - (voice_con+data_con)) *
  (lambda_for_voice/M_Ter);
25 IF (new_call_thr(voice_con,data_con) <= NO_CH- Trans_rate_thr)
  FROM EXTERN TO data_con RATE (M_Ter - (voice_con+data_con)) *
  (lambda_for_data/M_Ter);
//-----
26 FROM voice_with_queue[on] TO voice_with_queue[off] RATE lambda_for_voice;
27 FROM voice_with_queue[off] TO voice_with_queue[on] RATE mue_for_voice;
//-----NEW HANDOVER ARRIVAL-----
28 IF (Handover_thr(voice_con,data_con) <= NO_CH-1) AND (voice_with_queue == on)
  FROM EXTERN TO voice_con RATE mue_handover_for_voice;
29 IF (Handover_thr(voice_con,data_con) <= NO_CH-min_Trans_rate)
  FROM EXTERN TO data_con RATE mue_handover_for_data;
//-----
//Termination of Call and Outgoing Handover Request
30 FROM voice_con RATE voice_con*(mue_for_voice+mue_handover_for_voice);
31 FROM data_con RATE data_con*((1.0*Actual_rate_data(voice_con,data_con)
  /max_Trans_rate)*mue_for_data + mue_handover_for_data);
//-----RESULT PART-----
32 PRINT MEAN_voice := MEAN(voice_con);
33 PRINT MEAN_data := MEAN(data_con);
34 PRINT voice_blk := PROB(new_call_thr(voice_con,data_con) > NO_CH-1);
35 PRINT voice_hof := PROB(Handover_thr(voice_con,data_con) > NO_CH-1);
36 PRINT data_blk:= PROB(new_call_thr(voice_con,data_con)>NO_CH- Trans_rate_thr);
37 PRINT data_hof := PROB(Handover_thr(voice_con,data_con)>NO_CH- min_Trans_rate);
38 PRINT data_com := PROB(data_con > 0)
39 PRINT result := MEAN(reward);
40 PRINT av_data_rate := result/MEAN_data;
41 PRINT GOS_voice := voice_blk + voice_hof*10;
42 PRINT GOS_data := data_blk + data_hof*10;
43 PRINT utilization := MEAN(NO_of_occupied_CH)/NO_CH;
44 PRINT Throughput_data := MEAN_data * mue_for_data;
45 PRINT Throughput_voice := MEAN_voice * mue_for_voice;
46 PRINT Throughput := Throughput_data + Throughput_voice;

```

```

//-----PICTURE PART-----
47 PICTURE "Grade_of_service_voice"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Grade_of_service"
   CURVE GOS_voice
48 PICTURE "Grade_of_service_data"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Grade_of_service"
   CURVE GOS_data
49 PICTURE "utilization"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "utilization"
   CURVE utilization
50 PICTURE "AVG_DATA_RATE"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "AVG_DATA_RATE"
   CURVE av_data_rate
51 PICTURE "Blocking_voice"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Blocking"
   CURVE voice_blk
52 PICTURE "Blocking_data"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Blocking"
   CURVE data_blk
53 PICTURE "HO_Blocking_voice"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Blocking"
   CURVE voice_hof
54 PICTURE "HO_Blocking_data"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Blocking"
   CURVE data_hof
55 PICTURE "Throughput"
   PARAMETER lambda
   XLABEL "lambda"
   YLABEL "Throughput"
   CURVE Throughput

```

Figure 2: MOSEL-2 model for one cell cellular wireless networks

However, at higher threshold value 13 only good quality of service can be achieved at low rate. This can be explained as: when the threshold value is high then the number of voice connections who are blocked will be high, because the number of available channels in the cell will be less for the benefit of data connections and therefore, the grade-of-service will then increase. The minimum the threshold value, the better the number of voice calls who are being served, then the better the grade-of-service. An interesting result can be noticed in Figure 3, where at high traffic load, the minimum value of threshold (i.e. 4), the grade-of-service becomes the worst. This can be explained that very small value of threshold has a negative effect on both the new call blocking probability and the handover blocking probability. Different behavior can be noticed in Figure 4, which indicates that the data calls suffer more from the increase of the threshold value.




Figure 3: Grade-of-service against call rate for voice calls

This can be explained as: in Figure 3, at rate 0.20 and threshold value 10, the grade-of-service is less than 0.01. However, at the same rate and same threshold value in Figure 4, the grade-of-service is a round 0.09. From both Figures, it can be noticed that, the higher value of threshold has a negative effect on both new call blocking probability and handover blocking probability, which

leads to a negative effect on the grade-of-service. In Figure 5, the utilization of the cell is shown, where it can be noticed that higher value of threshold has extremely affect the utilization of the cell at different traffic load. This is clear by looking at Figure 5 (the curve where the threshold value is 13), following the same explanation above. The average transmission rate of data connection is shown in Figure 6. It is clear that as the traffic load increases, the average data transmission rate is also increases. However, for high threshold value (i.e. 13), the average transmission rate is better. This is clear as the higher threshold value, the higher the number of channels who are reserved for the benefit of data connection.




Figure 4: Grade-of-service against call rate for data calls




Figure 5: Cell utilization against call rate




Figure 6: Average data transmission rate against call rate

In Figures 7 and 8, the call blocking probability is shown for both the voice and data connection. Whereas in Figure 9 and 10, the handover blocking probability is shown for both voice and data. In our model, the ongoing handover calls are given higher priority on new calls for both voice and data connection. Therefore, it can be noticed from Figures 7 that the voice calls are only suffering when the threshold value is high for the benefit of the data connection. On the other hand, the data connection almost suffers at different traffic load as they consumes the channels faster than the voice calls, this can be noticed from Figure 7 and 8, where for example: at traffic load 0.50 and threshold value 13, the voice blocking probability reaches a value less than 0.10, whereas the data blocking probability reaches a value close to 0.70.

The suggested model gives significant results in Figures 9 and 10 for both voice and data calls. It can be noticed that the main objective of this analysis has been achieved. The handover blocking probability of both voice and data connections gives lower values at different traffic load,




Figure 7: voice blocking probability against call rate




Figure 8: data blocking probability against call rate

however, this can be significantly noticed at lower value of the threshold. This can be explained that when the threshold value is low, more channels will be reserved in the cell, but only for the benefit of handover calls. At this lower rate of the threshold, the throughput of the cell will be increased significantly at different traffic load. This last behavior can be noticed by looking at Figure 11.




Figure 9: voice handover blocking against call rate




Figure 10: data handover blocking against call rate




Figure 11: cell Throughput against call rate

4. CONCLUSIONS AND FUTURE WORK

The main objective of this work is to show the effectiveness of MOSEL-2 simulation in the performance modeling of cellular wireless networks by introducing an example. Firstly, the main features and environment of MOSEL-2 is presented. Secondly, the suggested model in this paper has been solved using MOSEL-2 language. The numerical results show two things: firstly, the effectiveness of MOSEL-2 in performance and modeling of cellular wireless networks in a very nice and friendly environment and secondly, the suggested model and its

call admission control algorithm are improving the handover blocking over the new call blocking for both voice and data connections. For model verification, it is shown that the model evaluation by MOSEL-2 takes less time than the model evaluation by the old version of MOSEL. In the future work, the work of this paper can be extended to show the efficiency of MOSEL-2 in the performance modeling and evaluation of different models of 2G (GSM/GPRS), 3G (UMTS), and even 4G (WiMAX/LTE) of mobile networks. Additionally, to improve the handover blocking probability over the new call blocking probability, a reservation policy of channels for handover calls can be suggested. Moreover, a queuing principle of the connection can also be suggested, where instead of directly blocking the connection when there is no free channels, the connection can wait for some time until the channel is available. This last suggestion will significantly improve both the new call blocking probability and the handover blocking probability.

REFERENCES

- Smida, B., V. Sampath and P. Marinier, May 2002. "Capacity degradation due to coexistence between second generation and 3G/WCDMA systems," in Proc. IEEE Veh. Technol. Conference, vol. 1, pp. 95-99.
- Heiska K., H. Posti, P. Muszynski, P. Aikio, J, Jan. 2002. Numminen and M. Hamalainen. "Capacity Reduction of WCDMA Downlink in the Presence of Interference From Adjacent Narrow-Band System," IEEE Trans. on Veh. Technol., vol. 51, Issue 1, pp. 37-51.
- Al-Begain, K., G., Bolch, H., Herold, , 2001. Practical Performance Modelling, Application of the MOSEL Language; Kluwer Academic Publishers, 409 pages.
- Wuechner P., 2003. Performance Modeling of mobile networks using MOSEL-2, M.S. Thesis, Department of computer science, University of Erlangen, Germany.
- Zreikat A. I., G. Bolch, 2007. "Performance Evaluation of Queuing Networks with Finite Capacity and non-exponential Distribution using MOSEL-2", Published in AMSE (Association for Modelling and Simulation in the Enterprises), Vol. 12, No. 4.
- Al-Begain K., J. Barner, G. Bolch, A. I. Zreikat, 2003. The Performance and reliability modeling language MOSEL and its applications, International Journal in Simulation: Systems, Science and technology, Vol. 3, No. 3-4, pp. 66-80.
- Zreikat A. I., G. Bolch, J. Sztrik, 2003. Performance Modeling of Non-Homogeneous Unreliable Multi-Server systems using MOSEL, International Journal in Computers and Mathematics with Applications, Elsevier, Vol. 46, pp. 293-312.
- Wüchner P., de Meer H., Bolch G., Roszik J., Sztrik J., 2007. Modeling Finite-Source Retrial Queueing Systems with Unreliable Heterogeneous Servers and Different Service Policies Using MOSEL, Proc. of ASMTA 2007 Conference, Prague, Czech Republic, June 4-6.
- Gunter B., J. Roszik, J. Sztrik, P. Wüchner, November 2006. Modeling Finite-Source Retrial Queueing Systems with Unreliable Heterogeneous Servers and Different Service Policies Using MOSEL. Technical Report MIP-0611, University of Passau, Germany.
- Wüchner P., H. de Meer, J. Barner, G. Bolch, 2005. MOSEL-2 - A Compact But Versatile Model Description Language And Its Evaluation Environment. Proc. of MMBnet'05 Workshop, University of Hamburg, Germany, September 8-9, pp. 51-59.
- Zreikat A. I., S., Yerima, K., Al-Begain January, 2008. "Performance Evaluation and Resource Management of Heirarchical MACRO-/MICRO Cellular Networks Using MOSEL-2", Published in Wireless Personal Communications, Springer (USA), and Vol. 44, No. 2.
- Wüchner P., K. Al-Begain, J. Barner, G. Bolch, May 2004. Modelling a single GSM/GPRS cell with delay tolerant voice calls using MOSEL-2, Proc. of UK Simulation Conference, Oxford, UK, pp. 88-94.
- Barner, J. ; G. Bolch, 2003. MOSEL-2. Modeling, Specification and Evaluation Language, Revision 2 . In: Sanders, William (Hrsg.) : Proceedngs of the 13th International Conference on Modeling Techniques and Tools for Computer Performance Evaluation (Performance TOOLS 2003 Urbana-Champaign, Illinois, 2 - 5.9 2003)..
- Zreikat A. I., Sep., 2011. "A new WIMAX/WI-FI Interoperability model and its performance Evaluation", submitted to Wireless Personal Communications, Springer (USA).
- Buetel B. 2003. "Integration of the Petri Net Analysator TimeNET into the Model Analysis Environment MOSEL," Technical Report, University of Erlangen-Nürnberg.
- Wüchner P., 2005. "Extending the Interface Between the Modelling Language MOSEL and CSPL by Adding Simulation Constructs," Technical Report, University of Erlangen-Nürnberg.
- Zimmermann A., 2005. TimeNET 3.0, User Manual, TU Berlin, 2001, available at: <http://pdv.cs.tu-berlin.de/~timenet/TimeNET-UserManual30.ps.gz>.
- Bolch G., S. Greiner, H. De Meer, and K. Trivedi, 1998. "Queueing Networks and Markov Chains", John Wiley & Sons New York.

AUTHOR BIOGRAPHIES



AYMEN I. ZREIKAT is an Associate professor at the Information Technology Department, Mu'tah University, Jordan. He has obtained his B.Sc. in Computer Science from Yarmouk University, Jordan in 1990 and MSc in Computational Engineering from University of Erlangen, Germany in 2000. Additionally, he has obtained his Ph.D. from Bradford University, UK in 2003. In January, 2001, he has joint the Performance Modelling and Engineering Research Group at the Computing Department of Bradford University, UK. His area of research is in the Performance Evaluation and Resource Management of 3G Wireless Mobile Networks and beyond. He has published a set of international books, Journal and Conference papers in this field and he is responsible for reviewing a set of papers in this area of research in a very reputable Journals. Furthermore, he is a member of some national coordinating committees; (such as the coordinating committee for the management of Queen Rania Al-Abdullah center for Educational Technology, Jordan, April, 2009, also committee for Higher Education Accreditation Commission, May, 2009,...etc) and international organizations; i.e., MOSEL group. He has been appointed as an Assistant Dean at the Faculty of Science, Mutah University, Jordan, from 1/9/2008-1/9/2010. From 1/9/2010-1/9/2011, he has been appointed as the chairman of the IT in Mu'tah University. His e-mail address is: siayzrei@yahoo.com.